Cars, Interchanges, Traffic Counters, and some Pretty Darned Good Knot Invariants

More at ωεβ/APAI

Rozansky Overbay Ohtsuki van der Veen

Abstract. Reporting on joint work with Roland van der Veen, I'll tell you some stories about ρ_1, an easy to define, strong, fast to compute, homomorphic, and well-connected knot invariant. ρ_1 was first studied by Rozansky and Overbay [Ro1, Ro2, Ro3, Ov] and Ohtsuki [Oh2], it has far-reaching generalizations, it is elementary and dominated by the coloured Jones polynomial, and I wish I understood it.

Common misconception. Dominated, elementary $\Rightarrow$ lesser.

We seek strong, fast, homomorphic knot and tangle invariants.

Strong. Having a small "kernel".

Fast. Computable even for large knots (best: poly time).





Piccirillo

Gompf–Scharlemann–Thompson

Homomorphic. Extends to tangles and behaves under tangle operations; especially gluings and doublings:



Why care for "Homomorphic"? Theorem. A knot K is *ribbon* iff there exists a $2n$-component tangle T with skeleton as below such that $\tau(T)=K$ and where $\delta(T)=U$ is the *untangle*:



Hear more at ωεβ/AKT.

Acknowledgement. This work was supported by NSERC grant RGPIN-2018-04350 and by the Chu Family Foundation (NYC).



Jones: Formulas stay; interpretations change with time.



Formulas. Draw an n-crossing knot K as on the right: all crossings face up, and the edges are marked with a running index $k\in\{1,\dots,2n+1\}$ and with rotation numbers φ_k. Let A be the $(2n+1)\times(2n+1)$ matrix constructed by starting with the identity matrix I, and adding a 2×2 block for each crossing:



$\longrightarrow$

A	col $i+1$	col $j+1$
row i	$-T^s$	T^s-1
row j	0	-1

Let $G=(g_{\alpha\beta})=A^{-1}$. For the trefoil example, it is:

$$A=\begin{pmatrix} 1 & -T & 0 & 0 & T-1 & 0 & 0\\ 0 & 1 & -1 & 0 & 0 & 0 & 0\\ 0 & 0 & 1 & -T & 0 & 0 & T-1\\ 0 & 0 & 0 & 1 & -1 & 0 & 0\\ 0 & 0 & T-1 & 0 & 1 & -T & 0\\ 0 & 0 & 0 & 0 & 0 & 1 & -1\\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix},$$

$$G=\begin{pmatrix} 1 & T & 1 & T & 1 & T & 1\\ 0 & 1 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T^2}{T^2-T+1} & 1\\ 0 & 0 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T^2}{T^2-T+1} & 1\\ 0 & 0 & \frac{1-T}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & 1\\ 0 & 0 & \frac{1-T}{T^2-T+1} & -\frac{(T-1)T}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & 1\\ 0 & 0 & 0 & 0 & 0 & 1 & 1\\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

"The Green Function"



Burau Alexander Fox Wirtinger Blanchfield

Note. The Alexander polynomial Δ is given by

$$\Delta=T^{(-\varphi-w)/2}\det(A),\qquad \text{with } \varphi=\sum_k\varphi_k,\ w=\sum_c s.$$

Classical Topologists: This is boring. Yawn.

Formulas, continued. Finally, set

$$R_1(c):=s\left(g_{ji}\left(g_{j+1,j}+g_{j,j+1}-g_{ij}\right)-g_{ii}\left(g_{j,j+1}-1\right)-1/2\right)$$

$$\rho_1:=\Delta^2\left(\sum_c R_1(c)-\sum_k\varphi_k\left(g_{kk}-1/2\right)\right).$$

In our example $\rho_1=-T^2+2T-2+2T^{-1}-T^{-2}$.

Theorem. ρ_1 is a knot invariant. Proof: later.

Classical Topologists: Whiskey Tango Foxtrot?

Cars, Interchanges, and Traffic Counters. Cars always drive forward. When a car crosses over a bridge it goes through with (algebraic) probability $T^s\sim 1$, but falls off with probability $1-T^s\sim 0^*$. At the very end, cars fall off and disappear. See also [Jo, LTW].



Jones Lin Tian Wang

$p=1-T^s$





image credits: diamondtraffic.com

image credits: Dall-E

[*] In algebra $x\sim 0$ if for every y in the ideal generated by x, $1-y$ is invertible.

References.

[BV1] D. Bar-Natan and R. van der Veen, *A Polynomial Time Knot Polynomial,* Proc. Amer. Math. Soc. **147** (2019) 377–397, arXiv:1708.04853.

[BV2] D. Bar-Natan and R. van der Veen, *Perturbed Gaussian Generating Functions for Universal Knot Invariants,* arXiv:2109.02057.

[Dr] V. G. Drinfel'd, *Quantum Groups,* Proc. Int. Cong. Math., 798–820, Berkeley, 1986.

[Jo] V. F. R. Jones, *Hecke Algebra Representations of Braid Groups and Link Polynomials,* Annals Math., **126** (1987) 335-388.

[La] R. J. Lawrence, *Universal Link Invariants using Quantum Groups,* Proc. XVII Int. Conf. on Diff. Geom. Methods in Theor. Phys., Chester, England, August 1988. World Scientific (1989) 55–63.

[LTW] X-S. Lin, F. Tian, and Z. Wang, *Burau Representation and Random Walk on String Links,* Pac. J. Math., **182-2** (1998) 289–302, arXiv:q-alg/9605023.

[Oh1] T. Ohtsuki, *Quantum Invariants,* Series on Knots and Everything **29**, World Scientific 2002.

[Oh2] T. Ohtsuki, *On the 2–loop Polynomial of Knots,* Geom. Top. **11** (2007) 1357–1475.

[Ov] A. Overbay, *Perturbative Expansion of the Colored Jones Polynomial,* Ph.D. thesis, University of North Carolina, August 2013, ωεβ/Ov.

[Ro1] L. Rozansky, *A Contribution of the Trivial Flat Connection to the Jones Polynomial and Witten's Invariant of 3D Manifolds, I,* Comm. Math. Phys. **175-2** (1996) 275–296, arXiv:hep-th/9401061.

[Ro2] L. Rozansky, *The Universal R-Matrix, Burau Representation and the Melvin-Morton Expansion of the Colored Jones Polynomial,* Adv. Math. **134-1** (1998) 1–31, arXiv:q-alg/9604005.

[Ro3] L. Rozansky, *A Universal U(1)-RCC Invariant of Links and Rationality Conjecture,* arXiv:math/0201139.

[Sch] S. Schaveling, *Expansions of Quantum Group Invariants,* Ph.D. thesis, Universiteit Leiden, September 2020, ωεβ/Scha.

Preliminaries

This is Rho.nb of http://drorbn.net/oa22/ap.

```
Once[<< KnotTheory`; << Rot.m];
```

```
Loading KnotTheory` version
  of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
Loading Rot.m from http://drorbn.net/la22/ap
  to compute rotation numbers.
```

The Program

```
R1[s_, i_, j_] :=
  s (g[j,i] (g[j+,j] + g[j,j+] - g[i,j]) - g[i,i] (g[j,j+] - 1) - 1/2);
Z[K_] := Module[{Cs, φ, n, A, s, i, j, k, Δ, G, ρ1},
   {Cs, φ} = Rot[K]; n = Length[Cs];
   A = IdentityMatrix[2 n + 1];
   Cases[Cs, {s_, i_, j_} :>
     (A[[{i, j}, {i + 1, j + 1}]] += ( -T^s  T^s - 1 )
                                     (  0      -1    ))];
   Δ = T^((-Total[φ] - Total[Cs[[All,1]]])/2) Det[A];
   G = Inverse[A];
   ρ1 = Sum[R1 @@ Cs[[k]], {k, 1, n}] - Sum[φ[[k]] (g[k,k] - 1/2), {k, 1, 2 n}];
   Factor@
    {Δ, Δ^2 ρ1 /. α_+ :> α + 1 /. g[α_, β_] :> G[[α, β]]}];
```

The First Few Knots

```
TableForm[Table[Join[{K[[1]]_K[[2]]}, Z[K]],
  {K, AllKnots[{3, 6}]}], TableAlignments → Center]
```

3_1	$\frac{1-T+T^2}{T}$	$\frac{(-1+T)^2\,(1+T^2)}{T^2}$
4_1	$-\frac{1-3\,T+T^2}{T}$	0
5_1	$\frac{1-T+T^2-T^3+T^4}{T^2}$	$\frac{(-1+T)^2\,(1+T^2)\,(2+T^2+2\,T^4)}{T^4}$
5_2	$\frac{2-3\,T+2\,T^2}{T}$	$\frac{(-1+T)^2\,(5-4\,T+5\,T^2)}{T^2}$
6_1	$-\frac{(-2+T)\,(-1+2\,T)}{T}$	$\frac{(-1+T)^2\,(1-4\,T+T^2)}{T^2}$
6_2	$-\frac{1-3\,T+3\,T^2-3\,T^3+T^4}{T^2}$	$\frac{(-1+T)^2\,(1-4\,T+4\,T^2-4\,T^3+4\,T^4-4\,T^5+T^6)}{T^4}$
6_3	$\frac{1-3\,T+5\,T^2-3\,T^3+T^4}{T^2}$	0

$p = 1 - T^s$

Fast!



```
Timing@
 Z[GST48 = EPD[X[14,1], X̄[2,29], X[3,40], X[43,4], X̄[26,5], X[6,95],
    X[96,7], X[13,8], X̄[9,28], X[10,41], X[42,11], X̄[27,12], X[30,15],
    X̄[16,61], X̄[17,72], X̄[18,83], X[19,34], X̄[89,20], X̄[21,92],
    X̄[79,22], X̄[68,23], X̄[57,24], X̄[25,56], X[62,31], X[73,32],
    X[84,33], X̄[50,35], X[36,81], X[37,70], X[38,59], X̄[39,54], X[44,55],
    X[58,45], X[69,46], X[80,47], X[48,91], X[90,49], X[51,82], X[52,71],
    X[53,60], X̄[63,74], X̄[64,85], X̄[76,65], X̄[87,66], X̄[67,94],
    X̄[75,86], X̄[88,77], X̄[78,93]]]
```

$$\Big\{170.313,\ \Big\{-\frac{1}{T^8}\left(-1+2\,T-T^2-T^3+2\,T^4-T^5+T^8\right)\left(-1+T^3-2\,T^4+T^5+T^6-2\,T^7+T^8\right),\ \frac{1}{T^{16}}(-1+T)^2\,\big(5-18\,T+33\,T^2-32\,T^3+2\,T^4+42\,T^5-62\,T^6-8\,T^7+166\,T^8-242\,T^9+108\,T^{10}+132\,T^{11}-226\,T^{12}+148\,T^{13}-11\,T^{14}-36\,T^{15}-11\,T^{16}+148\,T^{17}-226\,T^{18}+132\,T^{19}+108\,T^{20}-242\,T^{21}+166\,T^{22}-8\,T^{23}-62\,T^{24}+42\,T^{25}+2\,T^{26}-32\,T^{27}+33\,T^{28}-18\,T^{29}+5\,T^{30}\big)\Big\}\Big\}$$

Strong!

```
{NumberOfKnots[{3, 12}],
 Length@
  Union@Table[Z[K], {K, AllKnots[{3, 12}]}],
 Length@
  Union@Table[{HOMFLYPT[K], Kh[K]},
    {K, AllKnots[{3, 12}]}]}
```

```
{2977, 2882, 2785}
```

So the pair (Δ, ρ_1) attains 2,882 distinct values on the 2,977 prime knots with up to 12 crossings (a deficit of 95), whereas the pair (HOMFLYPT, Khovanov Homology) attains only 2,785 distinct values on the same knots (a deficit of 192).



Hoste Ocneanu Millett Freyd Lickorish Yetter Przytycki Traczyk Khovanov

Theorem. The Green function $g_{\alpha\beta}$ is the reading of a traffic counter at β, if car traffic is injected at α (if $\alpha=\beta$, the counter is *after* the injection point).



Example.

$\sum_{p\geq 0}(1-T)^p = T^{-1}$

$$G=\begin{pmatrix}1 & T^{-1} & 1\\ 0 & T^{-1} & 1\\ 0 & 0 & 1\end{pmatrix}$$

Proof. Near a crossing c with sign s, incoming upper edge i and incoming lower edge j, both sides satisfy the *g-rules*:



$$g_{i\beta}=\delta_{i\beta}+T^s g_{i+1,\beta}+(1-T^s)g_{j+1,\beta},\qquad g_{j\beta}=\delta_{j\beta}+g_{j+1,\beta},$$

and always, $g_{\alpha,2n+1}=1$: use common sense and $AG=I$ $(=GA)$.

Bonus. Near c, both sides satisfy the further *g-rules*:

$$g_{\alpha i}=T^{-s}(g_{\alpha,i+1}-\delta_{\alpha,i+1}),\qquad g_{\alpha j}=g_{\alpha,j+1}-(1-T^s)g_{\alpha i}-\delta_{\alpha,j+1}.$$

Invariance of ρ_1. We start with the hardest, Reidemeister 3:



⇒ Overall traffic patterns are unaffected by Reid3!
⇒ Green's $g_{\alpha\beta}$ is unchanged by Reid3, provided the cars injection site α and the traffic counters β are away.
⇒ Only the contribution from the R_1 terms within the Reid3 move matters, and using g-rules the relevant $g_{\alpha\beta}$'s can be pushed outside of the Reid3 area:




```
δ_{i_,j_} := If[i === j, 1, 0];
gRules_{s_,i_,j_} :=
  {g_{iβ_} :→ δ_{iβ} + T^s g_{i⁺,β} + (1 - T^s) g_{j⁺,β}, g_{jβ_} :→ δ_{jβ} + g_{j⁺,β},
   g_{α_,i} :→ T^{-s} (g_{α,i⁺} - δ_{α,i⁺}),
   g_{α_,j} :→ g_{α,j⁺} - (1 - T^s) g_{αi} - δ_{α,j⁺}}
lhs = R₁[1, j, k] + R₁[1, i, k⁺] + R₁[1, i⁺, j⁺] //.
    gRules_{1,j,k} ⋃ gRules_{1,i,k⁺} ⋃ gRules_{1,i⁺,j⁺};
rhs = R₁[1, i, j] + R₁[1, i⁺, k] + R₁[1, j⁺, k⁺] //.
    gRules_{1,i,j} ⋃ gRules_{1,i⁺,k} ⋃ gRules_{1,j⁺,k⁺};
Simplify[lhs == rhs]
True
```

Next comes Reid1, where we use results from an earlier example:

```
R₁[1, 2, 1] - 1 (g₂₂ - 1/2) /. g_{α_,β_} :→ (1 T⁻¹ 1; 0 T⁻¹ 1; 0 0 1)⟦α, β⟧
```

$$\frac{1}{T^2}-\frac{1}{T}-\frac{-1+\frac{1}{T}}{T} = 0$$



Invariance under the other moves is proven similarly.

Wearing my Topology hat the formula for R_1, and even the idea to look for R_1, remain a complete mystery to me.



Wearing my Quantum Algebra hat, I spy a Heisenberg algebra $\mathbb{H}=A\langle p,x\rangle/([p,x]=1)$:



$$\text{cars}\leftrightarrow p\qquad \text{traffic counters}\leftrightarrow x$$

Where did it come from? Consider $\mathfrak{g}_\epsilon := sl_{2+}^\epsilon := L\langle y,b,a,x\rangle$ with relations

$$[b,x]=\epsilon x,\quad [b,y]=-\epsilon y,\quad [b,a]=0,$$
$$[a,x]=x,\quad [a,y]=-y,\quad [x,y]=b+\epsilon a.$$

At invertible ϵ, it is isomorphic to sl_2 plus a central factor, and it can be quantized à la Drinfel'd [Dr] much like sl_2 to get an algebra $QU=A\langle y,b,a,x\rangle$ subject to (with $q=e^{\hbar\epsilon}$):

$$[b,a]=0,\quad [b,x]=\epsilon x,\quad [b,y]=-\epsilon y,$$
$$[a,x]=x,\quad [a,y]=-y,\quad xy-qyx=\frac{1-e^{-\hbar(b+\epsilon a)}}{\hbar}.$$

Now QU has an R-matrix solving Yang-Baxter (meaning Reid3),

$$R=\sum_{m,n\geq 0}\frac{y^n b^m\otimes(\hbar a)^m(\hbar x)^n}{m![n]_q!},\qquad ([n]_q!\text{ is a "quantum factorial"})$$

and so it has an associated "universal quantum invariant" à la Lawrence and Ohtsuki [La, Oh1], $Z_\epsilon(K)\in QU$.

Now $QU\cong\mathcal{U}(\mathfrak{g}_\epsilon)$ (only as algebras!) and $\mathcal{U}(\mathfrak{g}_\epsilon)$ represents into $\mathbb{H}$ via

$$y\to -tp-\epsilon\cdot xp^2,\quad b\to t+\epsilon\cdot xp,\quad a\to xp,\quad x\to x,$$

(abstractly, $\mathfrak{g}_\epsilon$ acts on its Verma module

$$\mathcal{U}(\mathfrak{g}_\epsilon)/(\mathcal{U}(\mathfrak{g}_\epsilon)\langle y,a,b-\epsilon a-t\rangle)\cong\mathbb{Q}[x]$$

by differential operators, namely via $\mathbb{H}$), so R can be pushed to $\mathcal{R}\in\mathbb{H}\otimes\mathbb{H}$.

Everything still makes sense at $\epsilon=0$ and can be expanded near $\epsilon=0$ resulting with $\mathcal{R}=\mathcal{R}_0(1+\epsilon\mathcal{R}_1+\cdots)$, with $\mathcal{R}_0=e^{t(xp\otimes 1-x\otimes p)}$ and $\mathcal{R}_1$ a quartic polynomial in p and x. So p's and x's get created along K and need to be pushed around to a standard location ("normal ordering"). This is done using

$$(p\otimes 1)\mathcal{R}_0=\mathcal{R}_0(T(p\otimes 1)+(1-T)(1\otimes p)),$$
$$(1\otimes p)\mathcal{R}_0=\mathcal{R}_0(1\otimes p),$$

and when the dust settles, we get our formulas for ρ_1. But QU is a quasi-triangular Hopf algebra, and hence **ρ_1 is homomorphic**. Read more at [BV1, BV2] and hear more at ωεβ/SolvApp, ωεβ/Dogma, ωεβ/DoPeGDO, ωεβ/FDA, ωεβ/AQDW.

Schaveling

Also, we can (and know how to) look at higher powers of ϵ and we can (and more or less know how to) replace sl_2 by arbitrary semi-simple Lie algebra (e.g., [Sch]). So **ρ_1 is not alone!**

These constructions are very similar to Rozansky-Overbay [Ro1, Ro2, Ro3, Ov] and hence to the "loop expansion" of the Kontsevich integral and the coloured Jones polynomial [Oh2].

If this all reads like **insanity** to you, it should (and you haven't seen half of it). Simple things should have simple explanations. Hence, **Homework.** Explain ρ_1 with no reference to quantum voodoo and find it a topology home (large enough to house generalizations!). Make explicit the homomorphic properties of ρ_1. Use them to do topology!

P.S. As a friend of Δ, ρ_1 gives a genus bound, sometimes better than Δ's. How much further does this friendship extend?

A Small-Print Page on ρ_d, $d>1$.

Definition. $\langle f(z_i), h(\zeta_i)\rangle_{\{z_i\}} := f(\partial_{\zeta_i})h\big|_{\zeta_i=0}$, so $\langle p^2x^2, \mathbb{e}^{g\pi\xi}\rangle = 2g^2$.

Baby Theorem. There exist (non unique) power series $r^{\pm}(p_1,p_2,x_1,x_2) = \sum_d \epsilon^d r^{\pm}_d(p_1,p_2,x_1,x_2) \in \mathbb{Q}[T^{\pm1},p_1,p_2,x_1,x_2][\![\epsilon]\!]$ with $\deg r^{\pm}_d \le 2d+2$ ("docile") such that the power series $Z^b = \sum \rho^b_d \epsilon^d :=$

$$\left\langle \exp\left(\sum_c r^s(p_i,p_j,x_i,x_j)\right), \exp\left(\sum_{\alpha,\beta} g_{\alpha\beta}\pi_\alpha\xi_\beta\right)\right\rangle_{\{p_\alpha,x_\beta\}}$$

is a bnot invariant. Beyond the once-and-for-all computation of $g_{\alpha\beta}$ (a matrix inversion), Z^b is computable in $O(n^d)$ operations in the ring $\mathbb{Q}[T^{\pm1}]$.

(Bnots are knot diagrams modulo the braid-like Reidemeister moves, but not the cyclic ones).

Theorem. There also exist docile power series $\gamma^\varphi(\bar p,\bar x) = \sum_d \epsilon^d\gamma^\varphi_d \in \mathbb{Q}[T^{\pm1},\bar p,\bar x][\![\epsilon]\!]$ such that the power series $Z = \sum \rho_d\epsilon^d :=$

$$\left\langle \exp\left(\sum_c r^s(p_i,p_j,x_i,x_j) + \sum_k \gamma^{\varphi_k}(\bar p_k,\bar x_k)\right),\right.$$
$$\left.\exp\left(\sum_{\alpha,\beta} g_{\alpha\beta}(\pi_\alpha+\bar\pi_\alpha)(\xi_\beta+\bar\xi_\beta) + \sum_\alpha \pi_\alpha\bar\xi_\alpha\right)\right\rangle_{\{p_\alpha,\bar p_\alpha,x_\beta,\bar x_\beta\}}$$

is a knot invariant, as easily computable as Z^b.

Implementation. Data, then program (with output using the Conway variable $z = \sqrt{T} - 1/\sqrt{T}$), and then a demo. See `Rho.nb` of ωεβ/ap.

```
V@γ_{1,φ_}[k_] = φ (1/2 - p̄_k x̄_k);  V@γ_{2,φ_}[k_] = -φ^2 p̄_k x̄_k / 2;
V@γ_{3,φ_}[k_] := -φ^3 p̄_k x̄_k / 6
```

```
V@r_{1,s_}[i_, j_] :=
 s (-1 + 2 p_i x_i - 2 p_j x_i + (-1 + T^s) p_i p_j x_i^2 + (1 - T^s) p_j^2 x_i^2 - 2 p_i p_j x_i x_j + 2 p_j^2 x_i x_j) / 2
```

```
V@r_{2,1}[i_, j_] :=
 (-6 p_i x_i + 6 p_j x_i - 3 (-1 + 3 T) p_i p_j x_i^2 + 3 (-1 + 3 T) p_j^2 x_i^2 + 4 (-1 + T) p_i^2 p_j x_i^3 -
   2 (-1 + T) (5 + T) p_i p_j^2 x_i^3 + 2 (-1 + T) (3 + T) p_j^3 x_i^3 + 18 p_i p_j x_i x_j -
   18 p_j^2 x_i x_j - 6 p_i^2 p_j x_i^2 x_j + 6 (2 + T) p_i p_j^2 x_i^2 x_j - 6 (1 + T) p_j^3 x_i^2 x_j -
   6 p_i p_j^2 x_i x_j^2 + 6 p_j^3 x_i x_j^2) / 12
```

```
V@r_{2,-1}[i_, j_] :=
 (-6 T^2 p_i x_i + 6 T^2 p_j x_i + 3 (-3 + T) T p_i p_j x_i^2 - 3 (-3 + T) T p_j^2 x_i^2 -
   4 (-1 + T) T p_i^2 p_j x_i^3 + 2 (-1 + T) (1 + 5 T) p_i p_j^2 x_i^3 - 2 (-1 + T) (1 + 3 T) p_j^3 x_i^3 +
   18 T^2 p_i p_j x_i x_j - 18 T^2 p_j^2 x_i x_j - 6 T^2 p_i^2 p_j x_i^2 x_j + 6 T (1 + 2 T) p_i p_j^2 x_i^2 x_j -
   6 T (1 + T) p_j^3 x_i^2 x_j - 6 T^2 p_i p_j^2 x_i x_j^2 + 6 T^2 p_j^3 x_i x_j^2) / (12 T^2)
```

```
V@r_{3,1}[i_, j_] :=
 (4 p_i x_i - 4 p_j x_i + 2 (5 + 7 T) p_i p_j x_i^2 - 2 (5 + 7 T) p_j^2 x_i^2 - 4 (-5 + 6 T) p_i^2 p_j x_i^3 +
   4 (-16 + 17 T + 2 T^2) p_i p_j^2 x_i^3 - 4 (-11 + 11 T + 2 T^2) p_j^3 x_i^3 + 3 (-1 + T) p_i^3 p_j x_i^4 -
   3 (-1 + T) (4 + 3 T) p_i^2 p_j^2 x_i^4 + (-1 + T) (13 + 22 T + T^2) p_i p_j^3 x_i^4 -
   (-1 + T) (4 + 13 T + T^2) p_j^4 x_i^4 - 28 p_i p_j x_i x_j + 28 p_j^2 x_i x_j + 36 p_i^2 p_j x_i^2 x_j -
   12 (9 + 2 T) p_i p_j^2 x_i^2 x_j + 24 (3 + T) p_j^3 x_i^2 x_j - 4 p_i^3 p_j x_i^3 x_j + 28 T p_i^2 p_j^2 x_i^3 x_j -
   4 (-6 + 17 T + T^2) p_i p_j^3 x_i^3 x_j + 4 (-5 + 10 T + T^2) p_j^4 x_i^3 x_j + 24 p_i p_j^2 x_i x_j^2 -
   24 p_j^3 x_i x_j^2 - 24 p_i^2 p_j^2 x_i^2 x_j^2 + 6 (10 + T) p_i p_j^3 x_i^2 x_j^2 - 6 (6 + T) p_j^4 x_i^2 x_j^2 -
   4 p_i p_j^3 x_i x_j^3 + 4 p_j^4 x_i x_j^3) / 24
```

```
V@r_{3,-1}[i_, j_] :=
 (-4 T^3 p_i x_i + 4 T^3 p_j x_i - 2 T^2 (7 + 5 T) p_i p_j x_i^2 + 2 T^2 (7 + 5 T) p_j^2 x_i^2 -
   4 T^2 (-6 + 5 T) p_i^2 p_j x_i^3 + 4 T (-2 - 17 T + 16 T^2) p_i p_j^2 x_i^3 -
   4 T (-2 - 11 T + 11 T^2) p_j^3 x_i^3 + 3 (-1 + T) T^2 p_i^3 p_j x_i^4 - 3 (-1 + T) T (3 + 4 T) p_i^2 p_j^2 x_i^4 +
   (-1 + T) (1 + 22 T + 13 T^2) p_i p_j^3 x_i^4 - (-1 + T) (1 + 13 T + 4 T^2) p_j^4 x_i^4 +
   28 T^3 p_i p_j x_i x_j - 28 T^3 p_j^2 x_i x_j - 36 T^3 p_i^2 p_j x_i^2 x_j + 12 T^2 (2 + 9 T) p_i p_j^2 x_i^2 x_j -
   24 T^2 (1 + 3 T) p_j^3 x_i^2 x_j + 4 T^3 p_i^3 p_j x_i^3 x_j - 28 T^2 p_i^2 p_j^2 x_i^3 x_j -
   4 T (-1 - 17 T + 6 T^2) p_i p_j^3 x_i^3 x_j + 4 T (-1 - 10 T + 5 T^2) p_j^4 x_i^3 x_j -
   24 T^3 p_i p_j^2 x_i x_j^2 + 24 T^3 p_j^3 x_i x_j^2 + 24 T^3 p_i^2 p_j^2 x_i^2 x_j^2 - 6 T^2 (1 + 10 T) p_i p_j^3 x_i^2 x_j^2 +
   6 T^2 (1 + 6 T) p_j^4 x_i^2 x_j^2 + 4 T^3 p_i p_j^3 x_i x_j^3 - 4 T^3 p_j^4 x_i x_j^3) / (24 T^3)
```

```
{p*, x*, p̄*, x̄*} = {π, ξ, π̄, ξ̄};  (z_i_)* := (z*)_i;
Zip_{}[𝓔_] := 𝓔;
Zip_{{z_, zs___}}[𝓔_] :=
 (Collect[𝓔 // Zip_{{zs}}, z] /. f_. z^d_. :> (D[f, {z*, d}])) /. z* → 0
```

```
gPair[fs_, w_] :=
 gPair[fs, w] =
  Collect[Zip_{Join@@Table[{p_α, p̄_α, x_α, x̄_α}, {α, w}]}[
    (Times @@ (V /@ fs))
     Exp[Sum[g_{α,β} (π_α + π̄_α) (ξ_β + ξ̄_β), {α, w}, {β, w}] - Sum[ξ̄_α π_α, {α, w}]]],
   g__, Factor]
```

```
T2z[p_] := Module[{q = Expand[p], n, c},
  If[q === 0, 0, c = Coefficient[q, T, n = Exponent[q, T]];
   c z^(2 n) + T2z[q - c (T^(1/2) - T^(-1/2))^(2 n)]]];
```

```
Z_d_[K_] := Module[{Cs, φ, n, A, s, i, j, k, Δ, G, d1, Z1, Z2, Z3},
  {Cs, φ} = Rot[K]; n = Length[Cs]; A = IdentityMatrix[2 n + 1];
  Cases[Cs, {s_, i_, j_} :> (A[[{i, j}, {i + 1, j + 1}]] += ( -T^s  T^s - 1 ))];
                                                              (  0     -1   )
  {Δ, G} = Factor@{T^((-Total[φ] - Total[Cs[[All, 1]]])/2) Det@A, Inverse@A};
  Z1 =
   Exp[Total[Cases[Cs, {s_, i_, j_} :> Sum[ϵ^d1 r_{d1,s}[i, j], {d1, d}]]] +
     Sum[ϵ^d1 γ_{d1,φ[[k]]}[k], {k, 2 n}, {d1, d}] /. γ_{_,0}[_] → 0];
  Z2 = Expand[F[{}, {}] × Normal@Series[Z1, {ϵ, 0, d}]] //.
    F[fs_, {es___}] × (f : (r | γ)_{ps__}[is__])^p_. :>
     F[Join[fs, Table[f, p]], DeleteDuplicates@{es, is}];
  Z3 = Expand[Z2 /. F[fs_, es_] :> Expand[gPair[
        Replace[fs, Thread[es → Range@Length@es], {2}], Length@es
       ] /. g_{α_,β_} :> G[[es[[α]], es[[β]]]] ]];
  Collect[{Δ, Z3 /. ϵ^p_. → p! Δ^(2 p) ϵ^p}, ϵ, T2z] ];
```

```
Z_2[GST48] (* takes a few minutes *)
```

$\{1 - 4z^2 - 61z^4 - 207z^6 - 296z^8 - 210z^{10} - 77z^{12} - 14z^{14} - z^{16},$
$1 + (38z^2 + 255z^4 + 1696z^6 + 16\,281z^8 + 86\,952z^{10} + 259\,994z^{12} + 487\,372z^{14} + 615\,066z^{16} + 543\,148z^{18} + 341\,714z^{20} +$
$153\,722z^{22} + 48\,983z^{24} + 10\,776z^{26} + 1554z^{28} + 132z^{30} + 5z^{32})\,\epsilon +$
$(-8 - 484z^2 + 9709z^4 + 165\,952z^6 + 1\,590\,491z^8 + 16\,256\,508z^{10} + 115\,341\,797z^{12} + 432\,685\,748z^{14} + 395\,838\,354z^{16} - 4\,017\,557\,792z^{18} - 23\,300\,064\,167z^{20} -$
$70\,082\,264\,972z^{22} - 142\,572\,271\,191z^{24} - 209\,475\,503\,700z^{26} - 221\,616\,295\,209z^{28} - 151\,502\,648\,428z^{30} - 23\,700\,199\,243z^{32} +$
$99\,462\,146\,328z^{34} + 164\,920\,463\,074z^{36} + 162\,550\,825\,432z^{38} + 119\,164\,552\,296z^{40} + 69\,153\,062\,608z^{42} + 32\,547\,596\,611z^{44} + 12\,541\,195\,448z^{46} +$
$3\,961\,384\,155z^{48} + 1\,021\,219\,696z^{50} + 212\,773\,106z^{52} + 35\,264\,208z^{54} + 4\,537\,548z^{56} + 436\,600z^{58} + 29\,536z^{60} + 1252z^{62} + 25z^{64})\,\epsilon^2\}$

```
TableForm[Table[Join[{K[[1]]_{K[[2]]}}, Z_3[K]], {K, AllKnots[{3, 6}]}], TableAlignments → Center] (* takes a few minutes *)
```

3_1	$1+z^2$	$1 + (2z^2+z^4)\epsilon + (2-4z^2+3z^4+4z^6+z^8)\epsilon^2 + (-12+74z^2-27z^4-20z^6+8z^8+6z^{10}+z^{12})\epsilon^3$
4_1	$1-z^2$	$1 + (-2+2z^4)\epsilon^2$
5_1	$1+3z^2+z^4$	$1 + (10z^2+21z^4+12z^6+2z^8)\epsilon + (6-28z^2+33z^4+364z^6+655z^8+536z^{10}+227z^{12}+48z^{14}+4z^{16})\epsilon^2 + (-60+970z^2+645z^4-3380z^6-3280z^8+7470z^{10}+19\,475z^{12}+20\,536z^{14}+12\,564z^{16}+4774z^{18}+1109z^{20}+144z^{22}+8z^{24})\epsilon^3$
5_2	$1+2z^2$	$1 + (6z^2+5z^4)\epsilon + (4-20z^2+43z^4+64z^6+26z^8)\epsilon^2 + (-36+498z^2-883z^4+100z^6+816z^8+556z^{10}+146z^{12})\epsilon^3$
6_1	$1-2z^2$	$1 + (-2z^2+z^4)\epsilon + (-4+4z^2+25z^4-8z^6+2z^8)\epsilon^2 + (12+154z^2-223z^4-608z^6+100z^8-52z^{10}+10z^{12})\epsilon^3$
6_2	$1-z^2-z^4$	$1 + (-2z^2-3z^4+2z^6+z^8)\epsilon + (-2-4z^2+29z^4+28z^6+42z^8-8z^{10}-2z^{12}+4z^{14}+z^{16})\epsilon^2 + (12+166z^2+155z^4-194z^6-2453z^8-1622z^{10}-1967z^{12}-258z^{14}+49z^{16}-30z^{18}+z^{20}+6z^{22}+z^{24})\epsilon^3$
6_3	$1+z^2+z^4$	$1 + (2+8z^2-16z^6-24z^8-16z^{10}-2z^{12})\epsilon^2$